

# PROTOCOL:

## Quantifying lipid droplet area in fat tissue stained for perilipin A

### Set up the file system

1. Make a "LD\_analysis" folder
2. Place the original images in LD\_analysis > Original Images
3. Make the following other folders in the LD\_analysis folder:
  - a. Thresholded Images, Corrected Images, Droplet Outlines

### Run the macro "DropletAreaPart1.ijm"

Note: When the macro starts, two dialog boxes will open asking you to choose a folder. For the first folder, select "Original Images." For the second folder, select "Thresholded Images."

```
run("Set Scale...", "distance=4 known=1 pixel=1 unit=um global");
run("Set Measurements...", "area display redirect=None decimal=2");


dirInput = getDirectory("Choose Directory ");
dirOutput = getDirectory("Choose Directory ");
list = getFileList(dirInput);

setBatchMode(true);
for (i=0;i<list.length;i++) {
    open(dirInput+ list[i]);

    run("8-bit");
    setAutoThreshold("Huang");
    run("Convert to Mask");
    run("Options...", "iterations=5 count=1 pad edm=8-bit do=Open");
    saveAs("Jpeg", dirOutput + File.nameWithoutExtension + "_Threshold");
    close();
}
```

### Manually draw in broken borders between lipid droplets

1. Open a thresholded image and its corresponding original image in ImageJ
2. Double click the paintbrush tool and select brush width "10" and color "white"
  - a. Brush width should approximately match the width of the lipid droplet membranes in the thresholded image
3. Use the paintbrush tool to draw in borders between lipid droplets, referencing the original image to determine whether there should be a border
4. Use the paintbrush tool to fill in areas that should not be counted as lipid droplets
5. Save the image to the "Corrected Images" folder



**Figure 1:** Examples of problems to correct manually. A) Torn membrane separating two lipid droplets should be joined with a line. B) A large, torn section of membrane is obscuring a lipid droplet, causing it to be detected as multiple small lipid droplets; the false “droplets” should be filled in.

### Run the macro “DropletAreaPart2.ijm”

Note: When the macro starts, two dialog boxes will open asking you to choose a folder. For the first folder, select “Corrected Images.” For the second folder, select “Droplet Outlines.”

```
run("Set Scale...", "distance=4 known=1 pixel=1 unit=um global");
run("Set Measurements...", "area display redirect=None decimal=2");

dirInput = getDirectory("Choose Directory ");
dirOutput = getDirectory("Choose Directory ");
list = getFileList(dirInput);

setBatchMode(true);
for (i=0;i<list.length;i++) {
  open(dirInput+ list[i]);

  setAutoThreshold("Huang");
  run("Convert to Mask");
  run("Analyze Particles...", "size=3-8000 circularity=0.30-1.00 show=Outlines display exclude summarize");

  saveAs("Jpeg", dirOutput + File.nameWithoutExtension + "_DropletArea");
  close();
}
```

### Manually check the outlined images for errors

If an error is found, manually draw in broken borders on the “Corrected Images” as before. When finished, delete all “Droplet Outlines” and rerun the part 2 macro.